
continually updated (i.e. updated grandfathering). In this paper we examine whether updated 

$TITLE  Updated grandfathering with entry and exit 
*       Date: 2008 --- K.E.Rosendahl and H. Storrøsten 
 
set      t       number of periods       /0*200/; 
 
alias (t,tt); 
 
scalars 
         C0      parameter cost function                 /88.868/, 
         C1      parameter cost function                 /0.56263/, 
         B       parameter demand function               /4/, 
         A       parameter demand function               /500/, 
         r       emission intensity BaU                  /1/, 
         my      growth in demand function               /0.02/, 
         delta   discount rate                           /0.1/, 
         k       year between emission and allocation    /5/, 
         m0      number of firms year 0                  /100/, 
         E0      percentage emission constraint year 0   /0.99/, 
         au      auction rate                            /0.5/, 
         gamma   annual change in emission constraint    /1/, 
         F       fixed annual cost                       /10/; 
 
parameter ET(t) Emission target; 
 
free variables 
         obj             objective, 
         op(t)           operating surplus in firm, 
         opold(t)        operating surplus in old firm, 
         opnew(t)        operating surplus in new firm, 
         npv(t)          net present value of operating surplus from per t; 
 
positive variables 
         e(t)            emission in firm, 
         q(t)            production in firm, 
         c(t)            production costs in firm, 
         p(t)            price, 
         s(t)            quota price, 
         LST             lump sum transfer to old firms, 
         I               investment cost, 
         m(t)            number of firms; 
 
equations 
         e_bau(t)        emission function bau, 
         e_eq(t)         emission constraint, 
         p_eq(t)         demand function, 
         q_eq(t)         production function, 
         c_eq(t)         cost function, 
         sP_eq(t)        quota price function: pure grandfathering, 
         sU1_eq(t)       quota price function: updating, 
         sU2_eq(t)       quota price function: updating last periods, 
         opnew_eq(t)     oper. surp. func. new firms: pure&upd. grandfathering, 
         opoldP_eq(t)    oper. surplus func. old firms: pure grandfathering, 
         opoldU1_eq(t)   oper. surplus func. old firms: updating, 
         opoldU2_eq(t)   oper. surplus func. old firms: updating first k years, 
         npvP_eq(t)      net present value function: pure grandfathering, 
         npvU_eq(t)      net present value function: updating, 
         LSTP_eq         lump sum transfer function: pure grandfathering, 
         LSTU_eq         lump sum transfer function: updating, 
         inv_bau(t)      investment condition in bau solution, 
         inv_eq(t)       investment condition, 
         inv2_eq(t)      investment condition no. 2, 
         close_eq(t)     closedown condition, 
         obj1_eq         objective function 
         ; 
 
* Sets BaU emissions to cost min. level 
e_bau(t)..       e(t) =e= r*q(t) ; 

* Total emissions target is allocated on all firms 
e_eq(t)..        e(t) =e= ET(t)/m(t) ; 
 
* Demand function: price is linear function of total production 
p_eq(t)..        p(t) =e= A - B*((1-my)**(ord(t)-1))*(m(t)*q(t)) ; 
 
* Price equals marg. prod. costs 
q_eq(t)..        p(t) =e= 2*C0*( q(t)*C1 + r*(r*q(t) - e(t)) ) ; 
 
* Production cost function 
c_eq(t)..        c(t) =e= C0*q(t)*q(t) * 
                 ( C1 + r**2 - 2*r*e(t)/q(t) +  (e(t)/q(t))**2 ) ; 
 
* Quota price equals marginal abatement costs (pure grandfathering) 
sP_eq(t)..        s(t) =e= -2*e(t)*C0 + (2*C0*A*r + 4*C0*C0*r*r*e(t)) / 
                 (2*C0*C1 + 2*C0*r*r + m(t)*B*((1-my)**(ord(t)-1))) ; 
 
* Relationship between quota price and MAC under updating 
sU1_eq(t)$(ord(t) le (card(t)-k)).. 
                 s(t) - ((1+delta)**(-k))*s(t+k)*(gamma**k)*(1-au) =e= 
                 -2*e(t)*C0 + (2*C0*A*r + 4*C0*C0*r*r*e(t)) / 
                 (2*C0*C1 + 2*C0*r*r + m(t)*B*((1-my)**(ord(t)-1))) ; 
 
* Quota price equals MAC last k periods under updating 
sU2_eq(t)$(ord(t) gt (card(t)-k)).. 
                 s(t) =e= -2*e(t)*C0 + (2*C0*A*r + 4*C0*C0*r*r*e(t)) / 
                 (2*C0*C1 + 2*C0*r*r + m(t)*B*((1-my)**(ord(t)-1))) ; 
 
* Oper. surplus for new firms under pure and updated grandfathering (and BaU) 
opnew_eq(t)..   opnew(t) =e= p(t)*q(t) - c(t) - e(t)*s(t) - F ; 
 
* Operating surplus for old firms under pure grandfathering (and BaU) 
opoldP_eq(t)..   opold(t) =e= p(t)*q(t) - c(t) - 
                 (e(t)-e("0")*ET(t)/ET("0")*(1-au))*s(t) - F ; 
 
* Operating surplus for old firms under updating except first k years 
opoldU1_eq(t)$(ord(t) gt k)..    opold(t) =e= p(t)*q(t) - c(t) - 
                 (e(t)-e(t-k)*(gamma**k)*(1-au))*s(t) - F ; 
 
* Operating surplus for old firms under updating first k years 
opoldU2_eq(t)$(ord(t) le k)..    opold(t) =e= p(t)*q(t) - c(t) - 
                 (e(t)-e("0")*ET(t)/ET("0")*(1-au))*s(t) - F ; 
 
* Net present value new firms under pure grandfathering (and BaU) 
npvP_eq(t)..     npv(t) =e= sum(tt$(ord(tt) ge ord(t)), 
                 (1+delta)**(ord(t)-ord(tt))*opnew(tt)) ; 
 
* Net present value new firms under updating 
npvU_eq(t)..     npv(t) =e= 
                 sum(tt$((ord(tt) ge ord(t)) and (ord(tt) lt (ord(t)+k))), 
                 (1+delta)**(ord(t)-ord(tt))*opnew(tt)) + 
                 sum(tt$(ord(tt) ge (ord(t)+k)), 
                 (1+delta)**(ord(t)-ord(tt))*opold(tt)) ; 
 
* Net present value of lump sum transfer under pure grandfathering 
LSTP_eq..        LST =e= sum(t, 
                 ((1+delta)**(-ord(t)))*e("0")*ET(t)/ET("0")*(1-au)*s(t)) ; 
 
* Net present value of lump sum transfer under updating 
LSTU_eq..        LST =e= sum(t$(ord(t) le k), 
                 ((1+delta)**(-ord(t)))*e("0")*ET(t)/ET("0")*(1-au)*s(t)) ; 
 
* Investment condition in BaU-scenario 
inv_bau(t)..      npv(t) =e= I ; 
 
* Investment condition when emission constraint is set 
* Either no investments (m unchanged) or investment costs equal npv 
inv_eq(t)$(ord(t) lt card(t))..  (npv(t+1) - I)*(m(t+1)-m(t)) =e= 0 ; 

* Make sure npv never exceeds investment costs 
inv2_eq(t)..     npv(t) =l= I ; 
 
* Make sure npv is never negative 
close_eq(t)..    npv(t) =g= 0 ; 
 
* Objective function - arbitrary objective variable 
obj1_eq..        obj =e= -sum(t,npv(t)*(1+delta)**(-(ord(t)-1))) ; 
 
model bau /q_eq, e_bau, p_eq, c_eq, npvP_eq, opnew_eq, inv_bau, obj1_eq/ ; 
 
model mfixed /e_eq, p_eq, q_eq, c_eq, sP_eq, npvP_eq, opnew_eq, 
                 opoldP_eq, obj1_eq/; 
 
model pure /e_eq, p_eq, q_eq, c_eq, sP_eq, opnew_eq, opoldP_eq, npvP_eq, 
                 LSTP_eq, inv_eq, inv2_eq, close_eq, obj1_eq/; 
 
model updated /e_eq, p_eq, q_eq, c_eq, sU1_eq, sU2_eq, opnew_eq, opoldU1_eq, 
                 LSTU_eq, opoldU2_eq, npvU_eq, inv_eq, inv2_eq, close_eq, 
obj1_eq/; 
 
q.lo(t) = 0.0001; 
e.lo(t) = 0.0001; 
p.lo(t) = 0.0001; 
m.lo(t) = 1; 
c.lo(t) = 0.0001; 
 
* Solve BaU model with no quota price and m(0)=100 to find investment cost 
* that is consistent with m(0)=100 
s.fx(t) = 0; 
m.fx("0") = 100; 
 
* Set an arbitrary value to initialize ET(t) before BaU solution 
ET(t)=100; 
 
solve bau using nlp minimizing obj; 
 
* Fix investment cost 
I.fx = I.l; 
 
* Set emissions constraint 
ET(t) = m.l("0")*e.l("0")*E0*(gamma**(ord(t)-1)) ; 
 
* Make the quota price free 
s.lo(t) = 0.0001; 
s.up(t) = inf; 
 
* Run first pure grandfathering with m fixed over entire horizon to check 
m.fx(t) = 100; 
 
solve mfixed using nlp minimizing obj; 
 
* Run pure grandfathering solution with m(0) fixed at 100, 
* but endogenous afterwards 
m.lo(t) = 1; 
m.up(t) = inf; 
m.fx("0") = 100; 
 
solve pure using nlp minimizing obj; 
 
* Run updated grandfathering solution 
solve updated using nlp minimizing obj; 

